wfopen  "C:\Users\Maryam\Desktop\BS Studies\PhD Thesis-II\EViews and STATA Progarm Codes\Chapter-8\Philippines.wf1"
'******************************************************************
'Group Plots for Real Exchange Rate (NT) & Productivity Gap
'******************************************************************
group gA rer_def_NT A_tilde
freeze(group_plot) gA.line(x)
group_plot.setelem(1) lcolor(black) symbol(7) lpat(1)
group_plot.setelem(2) lcolor(black) symbol(4) lpat(1)
group_plot.setelem(3) lcolor(black) symbol(1) lpat(1)
group_plot.setelem(3) lcolor(black)
group_plot.options linepat
group_plot.addtext(t) Real Exchange Rate (NT-based) and Productivity Gaps (Philippines & U.S): 1971-2013
group_plot.addtext(b) Year
group_plot.addtext(l) PNT  
group_plot.addtext(l) NER 
group_plot.addtext(l) rer_def_NT
group_plot.addtext(r) A_TILDE
'************************************************************
'************************************************************

create y 1971 2013
'importing data from Excel for Philippines
import  "C:\Users\Maryam\Desktop\BS Studies\PhD Thesis-II\EViews and STATA Progarm Codes\Chapter-8\Chapter 8.xlsx" range="Philippines"

'**************************************************************************************************
'VERIFYING THE ASSUMPTION OF PPP FOR Philippines AND U.S. TRADABLES PRICES
'**************************************************************************************************
'*************************************************************
'STEP 0: Tests for Unit Root in Individual Time Series
'*************************************************************

'***************************************************************
'Graph for Philippines's  pT
'***************************************************************
                                        
genr pT = pT
freeze(figure_pT) pT.line
figure_pT.addtext(t) pT (Philippines):  1971-2013
figure_pT.addtext(b) Year
figure_pT.addtext(l) pT
figure_pT.legend(off)
                                                 
'We see from the FIGURE that pT has a time trend to it.  So we would include both an intercept and a time trend in our unit root regression equtions. 

'*********************************************************************************
'ADF Unit Root Test for Philippines's pT
'We now run our first ADF test
'**********************************************************************************
 
freeze(table_8_2_pT_adf) pT.uroot(adf,trend,lag=1)

'Note that I specify lags, p = 1, for obtaining white noise residuals. The unit root test produces a t-value of -0.64 which is greater than our 5% criterion -3.52.  Thus, at this point, we cannot reject the null of a unit root.

'Now, let's check for white noise.  To do that, I first set all the residuals = 0, then run the adf test and finally will check for white noise.

genr resid = 0
freeze(mode=overwrite,pT_adf) pT.uroot(adf,const,trend,lag=1)
freeze(pT_adf_correl) resid.correl
 
'Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the pT series is not level stationary.

'The next thing I do is test whether the differenced series is stationary using the ADF test.  I once again begin by graphing the (differenced) series.
 
genr pTdiff = d(pT)
freeze(figure_pTdiff) pTdiff.line
figure_pTdiff.addtext(t) DpT (Philippines):  1971-2013
figure_pTdiff.addtext(b) Year
figure_pTdiff.addtext(l) DpT
figure_pTdiff.legend(off)

'The graph is not particularly illuminating.  Depending on how you look at it, it could have a time trend to it.  However, since the previous ADF test concluded that the equation for pT do have a time trend, we run the ADF test with both a constant as well as a time trend.

'So we begin the whole process over again: 

genr pTdiff = d(pT)
freeze(table_8_2_pTdiff1_adf) pTdiff.uroot(adf,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p =0.  The unit root test produces a t-value of -4.12 which is smaller our 5% criterion -2.93.  Thus, we may now reject the null of nonstationarity in first differenced series of pT.  There is no reason to go further.  The last thing we do is check for white noise.

genr resid = 0
freeze(mode=overwrite,pTdiff1_adf) pTdiff.uroot(adf,const,trend,info=sic)
freeze(pTdiff1_adf_correl) resid.correl

''Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the pT series is I(1).

'*********************************************************************************
'DF-GLS Unit Root Test for Philippines's pT
'We now run our first dfgls test
'**********************************************************************************
 
freeze(table_8_2_pT_dfgls) pT.uroot(dfgls,trend,info=sic)

'Note that I specify lags, p = 1, for obtaining white noise residuals. The unit root test produces a t-value of -0.79 which is greater than our 5% criterion -3.19.  Thus, at this point, we may not reject the null of a unit root.

'Now let's see if the series is difference stationary or not

genr pTdiff = d(pT)
freeze(table_8_2_pTdiff1_dfgls) pTdiff.uroot(dfgls,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p = 0.  The unit root test produces a t-value of -4.17 which is smaller than our 5% criterion -1.95.  Thus, we may now reject the null of nonstationarity in first differenced series of pT.  
 
''Putting it all together, I conclude that the pT series is I(1), a finding compatable with my ADF test results.

'***************************************************************
'Graph for Philippines's  pT_us_PPP
'***************************************************************
                                        
genr pT_us_PPP = pT_us_PPP
freeze(figure_pT_us_PPP) pT_us_PPP.line
figure_pT_us_PPP.addtext(t) pT_us_PPP (Philippines):  1971-2013
figure_pT_us_PPP.addtext(b) Year
figure_pT_us_PPP.addtext(l) pT_us_PPP
figure_pT_us_PPP.legend(off)
                                                 
'We see from the FIGURE that pT_us_PPP has a time trend to it.  So we would include both an intercepT_us_PPP and a time trend in our unit root regression equtions. 

'*********************************************************************************
'ADF Unit Root Test for Philippines's pT_us_PPP
'We now run our first ADF test
'**********************************************************************************
 
freeze(table_8_2_pT_us_PPP_adf) pT_us_PPP.uroot(adf,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 0.  The unit root test produces a t-value of 0.01 which is greater than our 5% criterion -3.52.  Thus, at this point, we cannot reject the null of a unit root.

'Now, let's check for white noise.  To do that, I first set all the residuals = 0, then run the adf test and finally will check for white noise.

genr resid = 0
freeze(mode=overwrite,pT_us_PPP_adf) pT_us_PPP.uroot(adf,const,trend,info=sic)
freeze(pT_us_PPP_adf_correl) resid.correl
 
'Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the pT_us_PPP series is not level stationary.

'The next thing I do is test whether the differenced series is stationary using the ADF test.  I once again begin by graphing the (differenced) series.
 
genr pT_us_PPPdiff = d(pT_us_PPP)
freeze(figure_pT_us_PPPdiff) pT_us_PPPdiff.line
figure_pT_us_PPPdiff.addtext(t) DpT_us_PPP (Philippines):  1971-2013
figure_pT_us_PPPdiff.addtext(b) Year
figure_pT_us_PPPdiff.addtext(l) DpT_us_PPP
figure_pT_us_PPPdiff.legend(off)

'The graph is not particularly illuminating.  Depending on how you look at it, it could have a time trend to it.  However, since the previous ADF test concluded that the equation for pT_us_PPP do have a time trend, we run the ADF test with both a constant as well as a time trend.

'So we begin the whole process over again: 

genr pT_us_PPPdiff = d(pT_us_PPP)
freeze(table_8_2_pT_us_PPPdiff1_adf) pT_us_PPPdiff.uroot(adf,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p =0.  The unit root test produces a t-value of -4.39 which is smaller our 5% criterion -2.93.  Thus, we may now reject the null of nonstationarity in first differenced series of pT_us_PPP.  There is no reason to go further.  The last thing we do is check for white noise.

genr resid = 0
freeze(mode=overwrite,pT_us_PPPdiff1_adf) pT_us_PPPdiff.uroot(adf,const,trend,info=sic)
freeze(pT_us_PPPdiff1_adf_correl) resid.correl

''Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the pT_us_PPP series is I(1).

'*********************************************************************************
'DF-GLS Unit Root Test for Philippines's pT_us_PPP
'We now run our first dfgls test
'**********************************************************************************
 
freeze(table_8_2a_pT_us_PPP_dfgls) pT_us_PPP.uroot(dfgls,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 1.  The unit root test produces a t-value of -0.88 which is greater than our 5% criterion -3.19.  Thus, at this point, we may not reject the null of a unit root.

'Now let's see if the series is difference stationary or not

genr pT_us_PPPdiff = d(pT_us_PPP)
freeze(table_8_2b_pT_us_PPPdiff1_dfgls) pT_us_PPPdiff.uroot(dfgls,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p = 0.  The unit root test produces a t-value of -4.43 which is smaller than our 5% criterion -1.95.  Thus, we may now reject the null of nonstationarity in first differenced series of pT_us_PPP.  
 
''Putting it all together, I conclude that the pT_us_PPP series is I(1), a finding compatable with my ADF test results.

''******************************************************************************************
''S1: Establishing Cointegration for Verifying PPP for Inter-Country Tradables Prices
'******************************************************************************************
'Now, by employing FMOLS and DOLS cointegration regression estimators, we shall calculate our LR cointegrating vectors.

equation table_8_2_LRPPP_fmols.cointreg(method=fmols) pT pT_us_PPP

equation table_8_2_LRPPP_dols.cointreg(method=dols, trend=constant, lag=1,lead=1 ) pT pT_us_PPP

'As proven by FMOLS and DOLS test results, valid cointegration holds between pT and pT_us_PPP. Thus, I can proceed with estimation of S2 condition. 

''************************************************************************************************
''S2: Testing for Equi-proportionate Relationship between Inter-Country Tradables Prices
'************************************************************************************************

freeze(table_8_2_LRPPP_fmols_Ftest) table_8_2_LRPPP_fmols.wald c(1)=1

freeze(table_8_2_LRPPP_dols_Ftest) table_8_2_LRPPP_dols.wald c(1)=1

'As proven by both Wald coefficient test results for FMOLS and DOLS estimates, PPP dos not validly hold for traded sector prices of Philippines and U.S. Thus, I will proceed with estimating the modified version of BS hypothesis. 

'***************************************************************************************
'ESTIMATING BALASSA-SAMUELSON EFFECT FOR rer_def_NT, rer_def_T  & a_tilde
'**************************************************************************************
'*************************************************************
'STEP 0: Tests for Unit Root in Individual Time Series
'*************************************************************
'***************************************************************
'Graph for Philippines's  rer_def_NT
'***************************************************************
                                        
genr rer_def_NT = rer_def_NT
freeze(figure_rer_def_NT) rer_def_NT.line
figure_rer_def_NT.addtext(t) rer_def_NT (Philippines):  1971-2013
figure_rer_def_NT.addtext(b) Year
figure_rer_def_NT.addtext(l) rer_def_NT
figure_rer_def_NT.legend(off)
                                                 
'We see from the FIGURE that rer_def_NT has a time trend to it.  So we would include both an intercept and a time trend in our unit root regression equtions. 

'********************************************************
'ADF Unit Root Test for Philippines's RER_DEF_NT
'********************************************************
 
freeze(table_8_6_rer_def_nt_adf) rer_def_nt.uroot(adf,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 0. The unit root test produces a t-value of -1.69 which is greater than our 5% criterion -3.52.  Thus, at this point, we cannot reject the null of a unit root.

'Now, let's check for white noise. To do that, I first set all the residuals = 0, then run the ADF test and finally will check for white noise.

genr resid = 0
freeze(mode=overwrite,rer_def_nt_adf) rer_def_nt.uroot(adf,const,trend,info=sic)
freeze(rer_def_nt_adf_correl) resid.correl
 
'Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the rer_def_nt series is not level stationary.

'The next thing I do is test whether the differenced series is stationary using the ADF test.  I once again begin by graphing the (differenced) series.
 
genr rer_def_ntdiff = d(rer_def_nt)
freeze(figure_rer_def_ntdiff) rer_def_ntdiff.line
figure_rer_def_ntdiff.addtext(t) drer_def_nt (Philippines):  1971-2013
figure_rer_def_ntdiff.addtext(b) Year
figure_rer_def_ntdiff.addtext(l) drer_def_nt
figure_rer_def_ntdiff.legend(off)

'From the graph, the series clearly does not have a time trend to it. So, I would test the series for unit with an intercept only.

'So we begin the whole process over again: 

genr rer_def_ntdiff = d(rer_def_nt)
freeze(table_8_6_rer_def_ntdiff1_adf) rer_def_ntdiff.uroot(adf,const,lag=2)

'Note that I specify lag p =1 to obtain white residuals.  The unit root test produces a t-value of -2.76 which is still greater than our 5% criterion -2.94.  Thus, we may not reject the null of non-stationarity in first differenced series of rer_def_nt.  There is no reason to go further.  The last thing we do is to check ADF regression result for white noise.

genr resid = 0
freeze(mode=overwrite,rer_def_ntdiff1_adf) rer_def_ntdiff.uroot(adf,const,lag=2)
freeze(rer_def_ntdiff1_adf_correl) resid.correl

''Based on the Q-statistic, I conclude that the residuals are white noise. Putting it all together, I conclude that the rer_def_nt series is greater than I(1).

'**************************************************************
'DF-GLS Unit Root Test for Philippines's RER_DEF_NT
'**************************************************************
 
freeze(table_8_6_rer_def_nt_dfgls) rer_def_nt.uroot(dfgls,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 0. The unit root test produces a t-value of -1.75 which is greater than our 5% criterion -3.19. Thus, at this point, we may not reject the null of a unit root. 

'Now let's see if the series is difference stationary or not

Genr rer_def_ntdiff = d(rer_def_nt)
freeze(table_8_6_rer_def_nt1diff1_dfgls) rer_def_ntdiff.uroot(dfgls,const,lag=2)

'Note that the SIC automatic lag selection picks no lags, p = 0.  The unit root test produces a t-value of -2.76 which is still greater than our 5% criterion -1.94. Thus, we may not reject the null of non-stationarity in first differenced series of rer_def_nt.  

''Putting it all together, I conclude that the rer_def_nt series is greater than I(1), a finding compatible with my ADF test results.

'*************************************************
'Graph for Philippines's Productivity (a_tilde)
'*************************************************
                                        
genr a_tilde = a_tilde
freeze(figurea_tilde) a_tilde.line
figurea_tilde.addtext(t) a_tilde (Philippines):  1971-2013
figurea_tilde.addtext(b) Year
figurea_tilde.addtext(l) a_tilde
figurea_tilde.legend(off)
                                                 
'We see from the FIGURE that a_tilde has time trend to it.  So we would include both an intercept and a time trend in our unit root regression equations. 

'*******************************************************
'ADF Unit Root Test for Philippines's Productivity
'*******************************************************
 
freeze(table_8_6_a_tilde_adf) a_tilde.uroot(adf,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p =0.  The unit root test produces a t-value of -2.35 which is greater than our 5% criterion -3.52.  Thus, at this point, we can reject the null of a unit root.

'Now, let's check for white noise.  To do that, I first set all the residuals = 0, then run the ADF test and finally will check for white noise.

genr resid = 0
freeze(mode=overwrite,a_tilde_adf) a_tilde.uroot(adf,const,trend,info=sic)
freeze(a_tilde_adf_correl) resid.correl
 
'Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the a_tilde series is not level stationary.

'The next thing I do is test whether the differenced series is stationary using the ADF test.  I once again begin by graphing the (differenced) series.
 
genr a_tildediff = d(a_tilde)
freeze(figure_a_tildediff) a_tildediff.line
figure_a_tildediff.addtext(t) da_tilde (Philippines):  1971-2013
figure_a_tildediff.addtext(b) Year
figure_a_tildediff.addtext(l) Da_tilde
figure_a_tildediff.legend(off)

'From the graph, the series clearly does not have a time trend to it. So, I would test the series for unit with an intercept only.

'So we begin the whole process over again: 

genr a_tildediff = d(a_tilde)
freeze(table_8_6_a_tildediff1_adf) a_tildediff.uroot(adf,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p =1.  The unit root test produces a t-value of -7.11 which is now smaller than our 5% criterion -2.94.  Thus, we may now reject the null of non-stationarity in first differenced series of a_tilde.  There is no reason to go further.  The last thing we do is to check ADF regression result for white noise.

genr resid = 0
freeze(mode=overwrite,a_tildediff1_adf) a_tildediff.uroot(adf,const,info=sic)
freeze(a_tildediff1_adf_correl) resid.correl

''Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the a_tilde series is I(1).

'************************************************************
'DF-GLS Unit Root Test for Philippines's Productivity
'************************************************************
 
freeze(table_8_6_a_tilde_dfgls) a_tilde.uroot(dfgls,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 0.  The unit root test produces a t-value of -2.45 which is greater than our 5% criterion -3.19.  Thus, at this point, we may not reject the null of a unit root.
 
'Now let's see if the series is difference stationary or not

genr a_tildediff = d(a_tilde)
freeze(table_8_6_a_tilde1diff1_dfgls) a_tildediff.uroot(dfgls,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p = 3.  The unit root test produces a t-value of -1.53 which is still greater than our 5% criterion -1.95. Thus, we may not reject the null of non-stationarity in first differenced series of a_tilde.  

''Putting it all together, I conclude that the a_tilde series is greater than I(1), a finding incompatible with my ADF test results.

'***************************************************************
'Graph for Philippines's  rer_def_T
'***************************************************************
                                        
genr rer_def_T = rer_def_T
freeze(figure_rer_def_T) rer_def_T.line
figure_rer_def_T.addtext(t) rer_def_T (Philippines):  1971-2013
figure_rer_def_T.addtext(b) Year
figure_rer_def_T.addtext(l) rer_def_T
figure_rer_def_T.legend(off)
                                                 
'We see from the FIGURE that rer_def_T has a time trend to it.  So we would include both an intercept and a time trend in our unit root regression equtions. 

'*********************************************************************************
'ADF Unit Root Test for Philippines's rer_def_T
'We now run our first ADF test
'**********************************************************************************
 
freeze(table_8_6_rer_def_T_adf) rer_def_T.uroot(adf,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p =0.  The unit root test produces a t-value of -2.68 which is greater than our 5% criterion -3.52.  Thus, at this point, we cannot reject the null of a unit root.

'Now, let's check for white noise.  To do that, I first set all the residuals = 0, then run the adf test and finally will check for white noise.

genr resid = 0
freeze(mode=overwrite,rer_def_T_adf) rer_def_T.uroot(adf,const,trend,info=sic)
freeze(rer_def_T_adf_correl) resid.correl
 
'Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the rer_def_T series is not level stationary.

'The next thing I do is test whether the differenced series is stationary using the ADF test.  I once again begin by graphing the (differenced) series.
 
genr rer_def_Tdiff = d(rer_def_T)
freeze(figure_rer_def_Tdiff) rer_def_Tdiff.line
figure_rer_def_Tdiff.addtext(t) Drer_def_T (Philippines):  1971-2013
figure_rer_def_Tdiff.addtext(b) Year
figure_rer_def_Tdiff.addtext(l) Drer_def_T
figure_rer_def_Tdiff.legend(off)

'The graph is not particularly illuminating.  Depending on how you look at it, it could have a time trend to it.  However, since the previous ADF test concluded that the equation for rer_def_T do have a time trend, we run the ADF test with both a constant as well as a time trend.

'So we begin the whole process over again: 

genr rer_def_Tdiff = d(rer_def_T)
freeze(table_8_6_rer_def_Tdiff1_adf) rer_def_Tdiff.uroot(adf,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p =0.  The unit root test produces a t-value of -6.57 which is smaller our 5% criterion -2.93.  Thus, we may now reject the null of nonstationarity in first differenced series of rer_def_T.  There is no reason to go further.  The last thing we do is check for white noise.

genr resid = 0
freeze(mode=overwrite,rer_def_Tdiff1_adf) rer_def_Tdiff.uroot(adf,const,trend,info=sic)
freeze(rer_def_Tdiff1_adf_correl) resid.correl

''Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the rer_def_T series is I(1).

'*********************************************************************************
'DF-GLS Unit Root Test for Philippines's rer_def_T
'We now run our first dfgls test
'**********************************************************************************
 
freeze(table_8_6a_rer_def_T_dfgls) rer_def_T.uroot(dfgls,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 0.  The unit root test produces a t-value of -2.57 which is greater than our 5% criterion -3.19.  Thus, at this point, we may not reject the null of a unit root.

'Now let's see if the series is difference stationary or not

genr rer_def_Tdiff = d(rer_def_T)
freeze(table_8_6b_rer_def_Tdiff1_dfgls) rer_def_Tdiff.uroot(dfgls,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p = 0.  The unit root test produces a t-value of -6.51 which is smaller than our 5% criterion -1.94.  Thus, we may now reject the null of nonstationarity in first differenced series of rer_def_T.  
 
''Putting it all together, I conclude that the rer_def_T series is I(1), a finding compatable with my ADF test results.

'*********************************************
'Single Equation Cointegration Methods
'*********************************************

'******************************************************************
''Graph the suspected cointegrated series together
'******************************************************************
'The first step is to print out a graph of the series.  This is very important!

group g1 rer_def_NT rer_def_T a_tilde
freeze(figure1) g1.line(x)
figure1.setelem(1) lcolor(black) symbol(1) lpat(1)
figure1.setelem(2) lcolor(black) symbol(4) lpat(1)
figure1.setelem(3) lcolor(black) symbol(7) lpat(1)
figure1.setelem(3) lcolor(black)
figure1.options linepat
figure1.addtext(t) rer_def_NT, lnpT & a_tilde (Philippines & U.S): 1971-2013
figure1.addtext(b) Year
figure1.addtext(l) rer_def_NT
figure1.addtext(l) lnpT
figure1.addtext(r) a_tilde

''*******************************************************
''S1.A.Engle-Granger Approach to Cointegration
'*******************************************************

genr resid = 0
equation eg.ls rer_def_NT c rer_def_T a_tilde
genr EC1 = resid

'First we test if the residuals of above regression are level stationary or not. If yes, next we'll proceed towards estimation of error correction model.

'***************************************************************
'Graph for Philippines's  EC
'***************************************************************                                               
genr EC1 = EC1
freeze(figure_EC1) EC1.line
figure_EC1.addtext(t) EC1 (Philippines):  1971-2013
figure_EC1.addtext(b) Year
figure_EC1.addtext(l) EC1
figure_EC1.legend(off)
                                                 
'We see from the FIGURE that EC has time trend to it.  So we would include both an intercept and trend in our unit root regression equtions. 

'*********************************************************************************
'EG Test for Cointegration
'**********************************************************************************
 
freeze(table_8_6_EGC) g1.coint(method=eg)

'The null hypothesis will be accepted as suggested by sample statistics.

''******************************************
''S1.B.Error Correction Model (ECM)
'*******************************************
''*******************************************************
'Selecting the number of lags in the VAR  *
'*******************************************************
'NOTE: We do this because we need to have the "right" number of lags when it comes time to estimate our VEC model and test for cointegration.

var var1.ls 1 6   g1
freeze(var1_lagtest1) var1.laglen(6)
freeze(var1_lagtest2) var1.testlags

'The laglength test above indicates that the should have VAR has 1 lags but the residuals were not white with this much number of lags. So, I had to raise the number of lags and on 2 lags  absolutely white residuals.

var var2.ls 1 2  g1
freeze(var2_artest1) var2.correl
freeze(var2_artest2) var2.qstats(12)
freeze(var2_artest3) var2.arlm(12)

'The residuals are not absolutely white noise byt I can't go beyond this number of lags.

'We now try different lags of d(rer_def_T) and d(a_tilde), comparing SIC values across specifications.

genr resid = 0
equation eg.ls rer_def_NT c rer_def_T a_tilde
genr ec1 = resid

var table_8_6_eg2a.ls 0 0 d(rer_def_NT)   @  c ec1(-1) d(rer_def_NT(-1)) d(rer_def_NT(-2)) 
var table_8_6_eg2b.ls 0 0 d(rer_def_NT)   @  c ec1(-1) d(rer_def_NT(-1)) d(rer_def_NT(-2))  d(rer_def_T(-1)) d(a_tilde(-1))

var table_8_6_eg2c.ls 0 0 d(rer_def_NT)   @  c ec1(-1) d(rer_def_NT(-1)) d(rer_def_NT(-2)) d(rer_def_T(-1)) d(rer_def_T(-2)) d(a_tilde(-1)) d(a_tilde(-2))

'The evidence suggests that Model A is best.  Now we test that model for serial correlation.

var table_8_6_eg2a.ls 0 0 d(rer_def_NT)   @   c ec1(-1) d(rer_def_NT(-1)) d(rer_def_NT(-2)) 
freeze(table_8_6_eg2a_artest1) table_8_6_eg2a.correl
freeze(table_8_6_eg2a_artest2) table_8_6_eg2a.qstats(12)
freeze(table_8_6_eg2a_artest3) table_8_6_eg2a.arlm(12)

'The residuals are not absolutely white noise. So let's try other two models.

var table_8_6_eg2b.ls 0 0 d(rer_def_NT)   @   c ec1(-1) d(rer_def_NT(-1)) d(rer_def_NT(-2)) d(rer_def_T(-1)) d(a_tilde(-1))
freeze(table_8_6_eg2b_artest1) table_8_6_eg2b.correl
freeze(table_8_6_eg2b_artest2) table_8_6_eg2b.qstats(12)
freeze(table_8_6_eg2b_artest3) table_8_6_eg2b.arlm(12)

var table_8_6_eg2c.ls 0 0 d(rer_def_NT)   @   c ec1(-1) d(rer_def_NT(-1)) d(rer_def_NT(-2)) d(rer_def_T(-1)) d(rer_def_T(-2)) d(a_tilde(-1)) d(a_tilde(-2))
freeze(table_8_6_eg2c_artest1) table_8_6_eg2c.correl
freeze(table_8_6_eg2c_artest2) table_8_6_eg2c.qstats(12)
freeze(table_8_6_eg2c_artest3) table_8_6_eg2c.arlm(12)

'Neither of the three models yield white noise residuals. However, Model C is performaing relatively better.

''*************************
''Estimating EC Model  
'**************************

'We'll now take the above specified model and turn it into an ECM. We shall run NW-HAC least squares model for establishing error correction mechanism.

'We now estimate the corresponding ECM:

equation table_8_6_ecm.ls(n) d(rer_def_NT) c ec1(-1) d(rer_def_NT(-1)) d(rer_def_NT(-2)) d(rer_def_T(-1)) d(rer_def_T(-2)) d(a_tilde(-1)) d(a_tilde(-2))

'Note that the SR effect is significant as the EC coefficient  (-0.20) is statistically significant at better than 10% significance level.

''**********************************************
''S2.A & S2.B: Obtaining LR Coefficients
'***********************************************
'Now, by employing FMOLS and DOLS cointegration regression estimators, finally we shall calculate our LR coefficient i.e. BS coefficient for Philippines against U.S.

equation table_8_6_LReqn1_fmols.cointreg(method=fmols) rer_def_NT rer_def_T a_tilde

equation table_8_6_LReqn2_dols.cointreg(method=dols, trend=constant, lag=2,lead=2 ) rer_def_NT rer_def_T a_tilde

'The BS coefficient obtained through FMOLS and DOLS estimators are 0.37 and 0.58. Thus, there is valid evidence in support of BS effect existing for Philippines.

''******************************************
'Multivariate Cointegration Approach
'******************************************

''*************************************************************
''Check if the VAR(2) model is dynamically stable
'*************************************************************
freeze(table_8_6_var2_varstable) var2.arroots(graph)

'The model is dynamically stable.

''**********************************************************************
''M1.A & M1.B: Identifying the number of cointegrating vectors
'***********************************************************************
'Having identified the appropriate number of lags to put in, I now go on to test for the appropriate number of cointegrating equations.

freeze(table_8_6_var2_coint) var2.coint(s,2)

'This command estimates all possible combinations of constants and trends in the level data series and the cointegrating equations. All the results indicate 1 cointegrating vectors.
'
'GENERAL NOTE:, in practice, cases 1 and 5 are rarely used. One should use case 1 only if one knows that all series have zero mean. Case 5 may provide a good fit in-sample but will produce implausible forecasts out-of-sample. As a rough guide, use case 2 if none of the series appear to have a trend. For trending series, use case 3 if you believe all trends are stochastic; if you believe some of the series are trend stationary, use case 4.

'Note that the 5 cases are identified under "Johansen cointegration test" in Eviews. They run from most restrictive (no constants in either the level series or CEs) to most general (trend terms in both the level series and CEs).

'CONCLUSION:  I conclude that rer_def_NT and a_tilde are not cointegrated in the Philippines's data.


